MEGA-BOARD USERS MANUAL

DISPLAY
TELECOMMUNICATIONS
CORPORATION

Corporate Headquarters:
4100 Spring Valley Road
Suite 400
Dallas, Texas 75234
(214) 991-1644

DISPLAY TELECOMMUNICATIONS CORPORATION OFFERS THE FOLLOWING PRODUCTS:

CASE WITH BEZEL	$ 99.95 each
MEGA-BOARD	$ 99.95 each
FULLY SOCKETED AND TESTED MEGA-BOARD	$199.95 each
ASSEMBLED AND TESTED MEGA-BOARD	$499.95 each
MEGA-BIOS ROM	$ 29.95 each
ROM LISTING	$ 19.95 each
USERS MANUAL	$ 19.95 each
KEYBOARD (PC COMPATIBLE)	$149.95 each
MAPPING PROM (TBP24S10)	$ 9.95 each
OCTAL TS SHIFTER (SN74LS322A)	$ 9.95 each
62 PIN PC CONNECTORS	$ 5.00 each
100NS DELAY LINE	$ 9.95 each
7NS DELAY LINE	$ 9.95 each
5 PIN DIN KEYBOARD CONNECTOR	$ 2.95 each
12 POSITION POWER CONNECTOR	$ 1.95 each
8 POSITION DIP SWITCH	$ 1.95 each
14.31818 CRYSTAL	$ 2.95 each
DUAL PERIPHERAL DRIVER (SN75477)	$ 2.95 each
TRIMMER CAPACITOR (JMC9613)	$ 1.95 each
64K MEMORY CHIPS	$ 5.00 each

BLANK BOARD REQUIREMENTS:
1 BOARD - (INCLUDES 2 BUS BARS)
1 MEGA-BIOS ROM
1 MAPPING PROM (TBP24S10)
1 OCTAL TS SHIFTER
7 62 PIN PC CONNECTORS
1 100NS DELAY LINE
1 7NS DELAY LINE
1 5 PIN DIN KEYBOARD CONNECTOR
1 12 POSITION POWER CONNECTOR
1 8 POSITION DIP SWITCH
1 14.31818 CRYSTAL
1 DUAL PERIPHERAL DRIVER (SN75477)
1 TRIMMER CAPACITOR (JMC9613)



Dip Switch Settings

SW2



DISPLAY
TELECOMMUNICATIONS
CORPORATION

Corporate Headquarters:
4100 Spring Valley Road
Suite 400
Dallas, Texas 75234
(214) 991-1644

Application Note 1

IBM MEMORY to INTEL HEX CONVERTER

The basic program that follows allows the user to convert blocks of memory in the PC to INTEL HEX format and store them in a file. That file can then be used to program EPROMS using any of the popular prom programming equipment, such as a GTEK 7128*, most of the DATA I/O programmers, or other devices.

This program facilitates patching the MEGA-BIOS, or other ROM. This can be accomplished by using DEBUG to move the data from ROM into RAM, making the modification, and writing it back out to a file.

For IBM PC and PC / XT compatible systems, the typical paragraph addresses are as follows:

ROM ADDRESS	LENGTH	PARAGRAPH ADDRESS	FUNCTION
F6000H	8192	F600	First BASIC ROM
F8000H	8192	F800	Second BASIC ROM
FA000H	8192	FA00	Third BASIC ROM
FC000H	8192	FC00	Fourth BASIC ROM
FE000H	8192	FE00	BIOS ROM

Each time a block is converted, a hex file is produced which starts at physical address 0000 and continues to an address in the target ROM equal to the starting address + (LENGTH - 1).

- NOTE -

Transferring, duplicating or copying copyrighted material, such as the IBM PC and IBM PC / XT ROMs may be a violation of the copyright laws. DisplayTel assumes no responsibility for the illegal use of this program. If in doubt, DON'T COPY IT!

* GTEK Incorporated
P.O. Box 289
Waveland, Mississippi 39576
(601) 467-8048

January 1984

```
1 REM "Display Telecommunications Corporation"
2 REM "   Memory to Intel Hex Converter"
3 REM "(C) Display Telecommunications Corp., 1984
4 REM "This program licensed for single machine use ONLY"
10 V$ = "1.0"
20 CLS
30 DEFINT A-Z
40 LOCATE 1,20
50 PRINT "ROM to Intel Hex Converter  ";V$
60 LOCATE 5,20
70 LINE INPUT "Enter Starting Paragraph . . ";ROM$
80 ROM$="&H"+ROM$
90 SEG = VAL (ROM$)
100 DEF SEG = SEG
110 LOCATE 7,20
120 LINE INPUT "File to Contain Hex Code? ";FILE$
130 OPEN "o",#1,FILE$
140 LOCATE 9,20
150 LINE INPUT "Enter Number of Bytes . . ";NUMBYTES$
160 NUMBYTES = VAL(NUMBYTES$)
170 NUMPARAGRAPHS = NUMBYTES / 16
180 LOCATE 11,20 : PRINT "Paragraphs to go . . . . ";
190 FOR J = 0 TO NUMPARAGRAPHS - 1
200 LOCATE 11,45
210 PRINT NUMPARAGRAPHS - (J+1)
220 PRINT #1,
230 PRINT #1,":10";
240 CKSUM = &H10
250 X = ((J AND &HFF0)/16) AND &HFF
260 CKSUM = CKSUM + X
270 PRINT #1,RIGHT$("00"+HEX$(X),2);
280 X = (J AND &HF) * 16
290 CKSUM = CKSUM + X
300 PRINT #1,RIGHT$("00"+HEX$(X),2);
310 PRINT #1,"00";
320 FOR I = 0 TO 15
330 X = PEEK (I+(J*16))
340 PRINT #1,RIGHT$("00"+HEX$(X),2);
350 CKSUM = CKSUM + X
360 NEXT I
370 CKSUM = CKSUM AND &HFF
380 CKSUM = -CKSUM AND &HFF
390 PRINT #1,RIGHT$("00"+HEX$(CKSUM),2);
400 NEXT J
410 PRINT #1, : PRINT #1,":00000001FF"
420 CLOSE
430 LOCATE 13,20 : LINE INPUT "More ROMs to do? ";Y$
440 IF LEFT$(Y$,1) = "y" OR LEFT$(Y$,1) = "Y" THEN 10
450 IF Y$="end" OR Y$="END" THEN END
460 CLS : SYSTEM
```